
FREEPASCAL GENERIC CONTAINER LIBRARY

(manual)

<http://code.google.com/p/stlpascal>

Contents

TVector	3
TStack	5
TDeque	7
TQueue	9
TLess, TGreater	11
TPriorityQueue	12
TArrayUtils	14
TOrderingArrayUtils	15
TSet	16
TMap	18
THashSet	21
THashMap	23

TVector

Implements selfresizing array. Indexing is 0-based.

Usage example:

```
uses gvector;  
  
type TVectorlli = specialize TVector<longint>;  
  
var Buffer: TVectorlli; i: longint;  
  
begin  
  Buffer := TVectorlli.Create;  
  {Push 5 elements at the end of array}  
  for i:=1 to 5 do  
    Buffer.PushBack(i);  
  {change 3rd element to 47}  
  Buffer[2] := 47;  
  {pop last element}  
  Buffer.PopBack;  
  {print all elements}  
  for i:=0 to Buffer.Size-1 do  
    writeln(Buffer[i]);  
  
  Buffer.Destroy;  
end.
```

Memory complexity: Uses at most 3times bigger memory than maximal array size (this is only case during reallocation). Normal consumption is at most twice as maximal array size.

Members list:

Method	Complexity guarantees
Description	
Create	O(1)
Constructor. Creates empty array.	
function Size(): SizeUInt	O(1)
Returns size of array.	
procedure PushBack(value: T)	Amortized O(1), some operations might take O(N) time, when array needs to be reallocated, but sequence of N operations takes O(N) time
Inserts at the end of array (increases size by 1)	

<code>procedure PopBack()</code>	$O(1)$
Removes element from the end of array (decreases size by 1). When array is empty, does nothing.	
<code>function IsEmpty(): boolean</code>	$O(1)$
Returns true when array is empty	
<code>procedure Insert(position: SizeUInt; value: T)</code>	$O(N)$
Inserts value at position. When position is greater than size, puts value at the end of array.	
<code>procedure Erase(position: SizeUInt; value: T)</code>	$O(N)$
Erases element from position. When position is outside of array does nothing.	
<code>procedure Clear</code>	$O(1)$
Clears array (set size to zero). But doesn't free memory used by array.	
<code>function Front: T</code>	$O(1)$
Returns first element from array.	
<code>function Back: T</code>	$O(1)$
Returns last element from array.	
<code>procedure Resize(num: SizeUInt)</code>	$O(N)$
Changes size of array to num. Doesn't guarantee anything about value of newly allocated elements.	
<code>procedure Reserve(num: SizeUInt)</code>	$O(N)$
Allocates at least num elements for array. Usefull when you want to pushback big number of elements and want to avoid frequent reallocation.	
<code>property item[i: SizeUInt]: T; default;</code>	$O(1)$
Property for accessing i-th element in array. Can be used just by square brackets (its default property).	
<code>property mutable[i: SizeUInt]: T;</code>	$O(1)$
Returns pointer to i-th element in array. Usefull when you store records.	

TStack

Implements stack.

Usage example:

```
uses gstack;  
  
type stacklli = specialize TStack<longint>;  
  
var data:stacklli; i:longint;  
  
begin  
  data:=stacklli.Create;  
  for i:=1 to 10 do  
    data.Push(10*i);  
  while not data.IsEmpty do begin  
    writeln(data.Top);  
    data.Pop;  
  end;  
  
  data.Destroy;  
end.
```

Memory complexity: Since underlying structure is TVector, memory complexity is same.

Members list:

Method	Complexity guarantees
Description	
Create	O(1)
Constructor. Creates empty stack.	
function Size(): SizeUInt	O(1)
Returns number of elements in stack.	
procedure Push(value: T)	Amortized O(1), some operations might take O(N) time, when array needs to be reallocated, but sequence of N operations takes O(N) time
Inserts element on the top of stack.	
procedure Pop()	O(1)
Removes element from the top of stack. If stack is empty does nothing.	
function IsEmpty(): boolean	O(1)
Returns true when stack is empty	

function Top: T	O(1)
Returns top element from stack.	

TDeque

Implements selfresizing array. Indexing is 0-based. Also implement constant time insertion from front.

Usage example:

```
uses gdeque;
```

```
type TDequelli = specialize TDeque<longint>;
```

```
var Buffer:TDequelli; i:longint;
```

```
begin
```

```
  Buffer := TDequelli.Create;
```

```
  {Push 5 elements at the end of array}
```

```
  for i:=1 to 5 do
```

```
    Buffer.PushBack(i);
```

```
  {change 3rd element to 47}
```

```
  Buffer[2] := 47;
```

```
  {pop last element}
```

```
  Buffer.PopBack;
```

```
  {push 3 element to front}
```

```
  for i:=1 to 3 do
```

```
    Buffer.PushFront(i*10);
```

```
  {print all elements}
```

```
  for i:=0 to Buffer.Size-1 do
```

```
    writeln(Buffer[i]);
```

```
  Buffer.Destroy;
```

```
end.
```

Memory complexity: Uses at most 3times bigger memory than maximal array size (this is only case during reallocation). Normal consumption is at most twice as maximal array size.

Members list:

Method	Complexity guarantees
Description	
Create	O(1)
Constructor. Creates empty array.	
function Size(): SizeUInt	O(1)
Returns size of array.	

<code>procedure PushBack(value: T)</code>	Amortized $O(1)$, some operations might take $O(N)$ time, when array needs to be reallocated, but sequence of N operations takes $O(N)$ time.
Inserts at the end of array (increases size by 1)	
<code>procedure PopBack()</code>	$O(1)$
Removes element from the end of array (decreases size by 1). When array is empty, does nothing.	
<code>procedure PushFront(value: T)</code>	Same as PushBack.
Inserts at the beginning of array (increases size by 1)	
<code>procedure PopFront()</code>	$O(1)$
Removes element from the beginning of array (decreases size by 1). When array is empty, does nothing.	
<code>function IsEmpty(): boolean</code>	$O(1)$
Returns true when array is empty	
<code>procedure Insert(position: SizeUInt; value: T)</code>	$O(N)$
Inserts value at position. When position is greater than size, puts value at the end of array.	
<code>procedure Erase(position: SizeUInt; value: T)</code>	$O(N)$
Erases element from position. When position is outside of array does nothing.	
<code>procedure Clear</code>	$O(1)$
Clears array (set size to zero). But doesn't free memory used by array.	
<code>function Front: T</code>	$O(1)$
Returns first element from array.	
<code>function Back: T</code>	$O(1)$
Returns last element from array.	
<code>procedure Resize(num: SizeUInt)</code>	$O(N)$
Changes size of array to num. Doesn't guarantee anything about value of newly allocated elements.	
<code>procedure Reserve(num: SizeUInt)</code>	$O(N)$
Allocates at least num elements for array. Useful when you want to pushback big number of elements and want to avoid frequent reallocation.	
<code>property item[i: SizeUInt]: T; default;</code>	$O(1)$
Property for accessing i-th element in array. Can be used just by square brackets (its default property).	
<code>property mutable[i: SizeUInt]: T;</code>	$O(1)$
Returns pointer to i-th element in array. Useful when you store records.	

TQueue

Implements queue.

Usage example:

```
uses gqueue;  
  
type queuelli = specialize TQueue<longint>;  
  
var data: queuelli; i: longint;  
  
begin  
  data := queuelli.Create;  
  for i := 1 to 10 do  
    data.Push(10*i);  
  while not data.IsEmpty do begin  
    writeln(data.Front);  
    data.Pop;  
  end;  
  
  data.Destroy;  
end.
```

Memory complexity: Since underlying structure is TDeque, memory complexity is same.

Members list:

Method	Complexity guarantees
Description	
Create	O(1)
Constructor. Creates empty queue.	
function Size(): SizeUInt	O(1)
Returns number of elements in queue.	
procedure Push(value: T)	Amortized O(1), some operations might take O(N) time, when array needs to be reallocated, but sequence of N operations takes O(N) time
Inserts element at the back of queue.	
procedure Pop()	O(1)
Removes element from the beginning of queue. If queue is empty does nothing.	
function IsEmpty(): boolean	O(1)
Returns true when queue is empty.	

<code>function Front: T</code>	<code>O(1)</code>
Returns the first element from queue.	

TLess, TGreater

Comparators classes. Can be used in PriorityQueue and Sorting as comparator functions. TLess is used for ordering from smallest element to largest, TGreater is used for oposite ordering.

TPriorityQueue

Implements priority queue. It's container which allow insertions of elements and then retrieval of the biggest one.

For specialization it needs two arguments. First is the type T of stored element. Second one is type comparator class, which should have class function `c(a,b: T):boolean` which return true, when a is strictly less than b (or in other words, a should be popped out after b).

Usage example:

```
{ $mode objfpc }  
  
uses gpriorityqueue;  
  
type  
  lesslli = class  
    public  
    class function c(a,b: longint):boolean; inline;  
  end;  
  
class function lesslli.c(a,b: longint):boolean; inline;  
begin  
  c:=a<b;  
end;  
  
type priorityqueuelli = specialize TPriorityQueue<longint, lesslli>;  
  
var data:priorityqueuelli; i:longint;  
  
begin  
  data:=priorityqueuelli.Create;  
  for i:=1 to 10 do  
    data.Push(random(1000));  
  while not data.IsEmpty do begin  
    writeln(data.Top);  
    data.Pop;  
  end;  
  
  data.Destroy;  
end.
```

Memory complexity: Since underlying structure is TVector, memory complexity is same.

Members list:

Method	Complexity guarantees
--------	-----------------------

Description	
Create	O(1)
Constructor. Creates empty priority queue.	
function Size(): SizeUInt	O(1)
Returns number of elements in priority queue.	
procedure Push(value: T)	Amortized $O(\lg N)$, some operations might take $O(N)$ time, when underlying array needs to be reallocated, but sequence of N operations takes $O(N \lg N)$ time.
Inserts element at the back of queue.	
procedure Pop()	$O(\lg N)$
Removes the biggest element from queue. If queue is empty does nothing.	
function IsEmpty(): boolean	O(1)
Returns true when queue is empty.	
function Top: T	O(1)
Returns the biggest element from queue.	

TArrayUtils

Set of utilities for manipulating arrays data.

Takes 3 arguments for specialization. First one is type of array (can be anything, which is accessible by [] operator, e. g. ordinary array, vector, ...), second one is type of array element.

Members list:

Method	Complexity guarantees
Description	
<code>procedure RandomShuffle(arr: TArray, size: SizeUInt)</code>	O(N)
Shuffles elements in array in random way	

TOrderingArrayUtils

Set of utilities for manipulating arrays data.

Takes 3 arguments for specialization. First one is type of array (can be anything, which is accessible by [] operator, e. g. ordinary array, vector, ...), second one is type of array element, third one is comparator class (see TPriorityQueue for definition of comparator class).

Usage example for sorting:

```
uses garrayutils , gutil , gvector ;

type vectorlli = specialize TVector<longint>;
    lesslli = specialize TLess<longint>;
    sortlli = specialize
        TOrderingArrayUtils<vectorlli , longint , lesslli >;

var data:vectorlli; n,i:longint;

begin
    randomize;
    data:=vectorlli.Create;
    read(n);
    for i:=1 to n do
        data.pushback(random(1000000000));
    sortlli.sort(data , data.size());
    for i:=1 to n do
        writeln(data[i-1]);

    data.Destroy;
end.
```

Members list:

Method	Complexity guarantees
Description	
<code>procedure Sort(arr: TArr, size:SizeUInt)</code>	O(N log N) average and worst case. Uses QuickSort, backed up by HeapSort, when QuickSort ends up in using too much recursion.
Sort array arr, with specified size. Array indexing should be 0 based.	
<code>function NextPermutation (arr: TArr, size:SizeUInt):boolean</code>	Worst case for one call $O(N)$. Going through all permutations takes $O(N!)$ time.
Orders elements on indexes $0, 1, \dots, size - 1$ into nearest lexicografic larger permutation.	

TSet

Implements container for storing ordered set of unique elements. Takes 2 arguments for specialization, first one is type of elements, second one is comparator class. Usage example:

```
uses gset, gutil;  
  
type lesslli=specialize TLess<longint>;  
    setlli=specialize TSet<longint, lesslli>;  
  
var data:setlli; i:longint; iterator:setlli.TIterator;  
  
begin  
    data:=setlli.Create;  
  
    for i:=0 to 10 do  
        data.insert(i);  
  
        {Iteration through elements}  
        iterator:=data.Min;  
        repeat  
            writeln(iterator.Data);  
        until not iterator.next;  
        {Don't forget to destroy iterator}  
        iterator.Destroy;  
  
        iterator := data.FindLess(7);  
        writeln(iterator.Data);  
        iterator.Destroy;  
  
    data.Destroy;  
end.
```

Memory complexity: Size of stored elements + constant overhead for each stored element (3 pointers + one boolean).

Members list:

Method	Complexity guarantees
Description	
Create	O(1)
Constructor. Creates empty set.	
function Size(): SizeUInt	O(1)
Returns number of elements in set.	

<code>procedure Insert(value: T)</code>	$O(\lg N)$, N is number of elements in set
Inserts element into set, if given element is already there nothing happens.	
<code>function InsertAndGetIterator (value: T):TIterator</code>	$O(\log_2 N)$
Inserts element into set, if given element is already there nothing happens. Also returns iterator pointing on given element.	
<code>procedure Delete(value: T)</code>	$O(\lg N)$, N is number of elements in set
Deletes value from set. If element is not in set, nothing happens.	
<code>function Find(value: T):TIterator</code>	$O(\lg N)$
Searches for value in set. If value is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from set.	
<code>function FindLess(value: T):TIterator</code>	$O(\lg N)$
Searches for greatest element less than value in set. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from set.	
<code>function FindLessEqual(value: T):TIterator</code>	$O(\lg N)$
Searches for greatest element less or equal than value in set. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from set.	
<code>function FindGreater(value: T):TIterator</code>	$O(\lg N)$
Searches for smallest element greater than value in set. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from set.	
<code>function FindGreaterEqual(value: T):TIterator</code>	$O(\lg N)$
Searches for smallest element greater or equal than value in set. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from set.	
<code>function Min:TIterator</code>	$O(\lg N)$
Returns iterator pointing to the smallest element of set. If set is empty returns nil.	
<code>function Max:TIterator</code>	$O(\lg N)$
Returns iterator pointing to the largest element of set. If set is empty returns nil.	
<code>function IsEmpty(): boolean</code>	$O(1)$
Returns true when set is empty.	

Some methods return type `TIterator`, which has following methods:

Method	Complexity guarantees
Description	
<code>function Next:boolean</code>	$O(\lg N)$ worst case, but traversal from smallest element to largest takes $O(N)$ time
Moves iterator to smallest larger element in set. Returns true on success. If the iterator is already pointing on largest element returns false.	
<code>function Prev:boolean</code>	$O(\lg N)$ worst case, but traversal from largest element to smallest takes $O(N)$ time
Moves iterator to largest smaller element in set. Returns true on success. If the iterator is already pointing on smallest element returns false.	
<code>property Data:T</code>	$O(1)$
Property, which allows reading of the element.	

TMap

Implements container for ordered associative array with unique keys. Takes 3 arguments for specialization, first one is type of keys, second one is type of values, third one is comparator class for keys. Usage example:

```
uses gmap, gutil;  
  
type lesslli=specialize TLess<longint>;  
    maplli=specialize TMap<longint, longint, lesslli>;  
  
var data:maplli; i:longint; iterator:maplli.TIterator;  
  
begin  
    data:=maplli.Create;  
  
    for i:=0 to 10 do  
        data[i]:=10*i;  
  
        writeln(data[7]);  
        data[7] := 42;  
  
        {Iteration through elements}  
        iterator:=data.Min;  
        repeat  
            writeln(iterator.Key, ' ', iterator.Value);  
            iterator.Value := 47;  
        until not iterator.next;  
        iterator.Destroy;  
  
        iterator := data.FindLess(7);  
        writeln(iterator.Value);  
        iterator.Destroy;  
  
        data.Destroy;  
end.
```

Memory complexity: Size of stored base + constant overhead for each stored element (3 pointers + one boolean).

Members list:

Method	Complexity guarantees
Description	
Create	O(1)

Constructor. Creates empty map.	
<code>function Size(): SizeUInt</code>	$O(1)$
Returns number of elements in map.	
<code>procedure Insert(key: TKey; value: TValue)</code>	$O(\lg N)$, N is number of elements in map
Inserts key value pair into map. If key was already there, it will have new value assigned.	
<code>function InsertAndGetIterator(key:TKey; value: TValue):TIterator</code>	$O(\log_2 N)$
Same as Insert but also returns iterator pointing to inserted element.	
<code>procedure Delete(key: TKey)</code>	$O(\lg N)$
Deletes key (and associated value) from map. If element is not in map, nothing happens.	
<code>function Find(key: T):TIterator</code>	$O(\lg N)$
Searches for key in map. If value is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from map.	
<code>function FindLess(key: T):TIterator</code>	$O(\lg N)$
Searches for greatest element less than key in map. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from map.	
<code>function FindLessEqual(key: T):TIterator</code>	$O(\lg N)$
Searches for greatest element less or equal than key in map. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from map.	
<code>function FindGreater(key: T):TIterator</code>	$O(\lg N)$
Searches for smallest element greater than key in map. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from map.	
<code>function FindGreaterEqual(key: T):TIterator</code>	$O(\lg N)$
Searches for smallest element greater or equal than key in map. If such element is not there returns nil. Otherwise returns iterator, which can be used for retrieving data from map.	
<code>function Min:TIterator</code>	$O(\lg N)$
Returns iterator pointing to smallest key of map. If map is empty returns nil.	
<code>function Max:TIterator</code>	$O(\lg N)$
Returns iterator pointing to largest key of map. If map is empty returns nil.	
<code>function IsEmpty(): boolean</code>	$O(1)$
Returns true when map is empty.	
<code>property item[i: Key]: TValue; default;</code>	$O(\ln N)$
Property for accessing key i in map. Can be used just by square brackets (its default property).	

Some methods return type `TIterator`, which has following methods:

Method	Complexity guarantees
Description	
<code>function Next:boolean</code>	$O(\lg N)$ worst case, but traversal from smallest element to largest takes $O(N)$ time
Moves iterator to element with smallest larger key. Returns true on success. If the iterator is already pointing on element with largest key returns false.	
<code>function Prev:boolean</code>	$O(\lg N)$ worst case, but traversal from largest element to smallest takes $O(N)$ time
Moves iterator to element with largest smaller key. Returns true on success. If the iterator is already pointing on element with smallest key returns false.	

<code>property Key:TKey</code>	$O(1)$
Property, which allows reading the key.	
<code>property Value:TValue</code>	$O(1)$
Property, which allows reading and writing of the value.	
<code>property MutableValue:PValue</code>	$O(1)$
Returns pointer on stored value. Usefull for accessing records and objects.	

THashSet

Implements container for storing unordered set of unique elements. Takes 2 arguments for specialization, first one is type of elements, second one is a hash functor (class which has class function hash, which takes element and number n and returns hash of the element in range $0, 1, \dots, n-1$). Usage example:

```
{ $mode objc }

uses ghashset;

type hashlli=class
    public
        class function hash(a:longint; b:SizeUInt):SizeUInt;
    end;
    setlli=specialize THashSet<longint, hashlli>;

class function hashlli.hash(a:longint; b:SizeUInt):SizeUInt;
begin
    hash:= a mod b;
end;

var data:setlli; i:longint; iterator:setlli.TIterator;

begin
    data:= setlli.Create;

    for i:=0 to 10 do
        data.insert(i);

        { Iteration through elements }
        iterator:=data.Iterator;
        repeat
            writeln(iterator.Data);
        until not iterator.Next;
        { Don't forget to destroy iterator }
        iterator.Destroy;

    data.Destroy;
end.
```

Memory complexity: Arounds two times of size of stored elements Members list:

Method	Complexity guarantees
--------	-----------------------

Description	
Create	O(1)
Constructor. Creates empty set.	
function Size(): SizeUInt	O(1)
Returns number of elements in set.	
procedure Insert(value: T)	O(1) on average
Inserts element into set, if given element is already there nothing happens.	
procedure Delete(value: T)	O(1) on average
Deletes value from set. If element is not in set, nothing happens.	
function Contains(value: T):boolean	O(1) on average
Checks whether element is in set.	
function Iterator:TIterator	O(1)
Returns iterator allowing traversal through set. If set is empty returns nil.	
function IsEmpty(): boolean	O(1)
Returns true when set is empty.	

Some methods return type TIterator, which has following methods:

Method	Complexity guarantees
Description	
function Next:boolean	O(N) worst case, but traversal of whole set takes O(N) time
Moves iterator to next larger element in set. Returns true on success. If the iterator is already pointing on last element returns false.	
property Data:T	O(1)
Property, which allows reading of the element.	

THashMap

Implements container for unordered associative array with unique keys. Takes 3 arguments for specialization, first one is type of keys, second one is type of values, third one is a hash functor (class which has class function hash, which takes element and number n and returns hash of the element in range $0, 1, \dots, n - 1$). Usage example:

```
{$mode objfpc}

uses ghashmap;

type hashlli=class
    public
        class function hash(a:longint; b:SizeUInt):SizeUInt;
    end;
    maplli=specialize THashMap<longint, longint, hashlli>;

class function hashlli.hash(a:longint; b:SizeUInt):SizeUInt;
begin
    hash:= a mod b;
end;

var data:maplli; i:longint; iterator:maplli.TIterator;

begin
    data:=maplli.Create;

    for i:=0 to 10 do
        data[i] := 17*i;

    data.delete(5);

    {Iteration through elements}
    iterator:=data.Iterator;
    repeat
        writeln(iterator.Key, ' ', iterator.Value);
    until not iterator.Next;
    {Don't forget to destroy iterator}
    iterator.Destroy;

    data.Destroy;
end.
```

Memory complexity: Arounds two times of size of stored elements

Members list:

Method	Complexity guarantees
Description	
Create	$O(1)$
Constructor. Creates empty map.	
function Size(): SizeUInt	$O(1)$
Returns number of elements in map.	
procedure Insert(key: TKey; value: TValue)	$O(1)$
Inserts key value pair into map. If key was already there, it will have new value assigned.	
procedure Delete(key: TKey)	$O(\lg N)$
Deletes key (and associated value) from map. If element is not in map, nothing happens.	
function Contains(key: TKey): boolean	$O(1)$ on average
Checks whether element with given key is in map.	
function Iterator: TIterator	$O(1)$ on average
Returns iterator allowing traversal through map. If map is empty returns nil.	
function IsEmpty(): boolean	$O(1)$
Returns true when map is empty.	
property item[i: Key]: TValue; default;	$O(1)$ on average
Property for accessing key i in map. Can be used just by square brackets (its default property).	

Some methods return type TIterator, which has following methods:

Method	Complexity guarantees
Description	
function Next: boolean	$O(N)$ worst case, but traversal of whole set takes $O(N)$ time
Moves iterator to next larger element in set. Returns true on success. If the iterator is already pointing on last element returns false.	
property Key: TKey	$O(1)$
Property, which allows reading the key.	
property Value: TValue	$O(1)$
Property, which allows reading and writing of the value.	
property MutableValue: PValue	$O(1)$
Returns pointer on stored value. Usefull for accessing records and objects.	